

大学入試問題（求積）の解法への Quantifier Elimination の応用について

傍田 裕子*・北本 卓也

Application of Quantifier Elimination
to University Entrance Examination (Quadrature)

SOBATA Yuko*, KITAMOTO Takuya

(Received September 25, 2020)

概要

計算機の急速な発展に伴い、その計算能力を活用する試みが様々な分野で行われている。教育分野においても数学の問題作成及び検証に数式処理を活用する実践及び検証がおこなわれつつある。本稿では、大学入試問題、特に「通過領域の問題」と呼ばれる問題を数式処理システムの QE ツールを用いて解く方法について述べる。数式処理システムは Mathematica, QE ツールは Mathematica の組み込みシンボルである Reduce を用いる。
キーワード：数式処理システム, QE, Mathematica

1 はじめに

1.1 Quantifier Elimination と大学入試問題

数式処理のアルゴリズムである限量子消去 (quantifier elimination: QE) は、 \exists (存在する), \forall (すべての) といった限量子がついた一階述語論理式を入力として、それと等価で限量子がない論理式を出力する。日本語による解説としては [1] が詳しく、同書では大学入試問題の解法も解説されている。国立大学 2 次試験の数学の問題はおよそ 4 割が実閉体の限量子消去 (RCF-QE) による演繹の適用範囲にあり [2], 2011 年に始まった人工知能プロジェクト「ロボットは東大に入れるか (略称: 東ロボ)」では、数学の解答システムの主ソルバーとして RCF-QE が用いられた [3]。

1.2 扱う問題と手法

以上を背景として、本稿では、大学入試問題のうちの求積の問題、特に通過領域の問題に焦点をあて、数式処理システムの QE ツールを用いて解く方法について述べる。

東ロボでは問題の入力から解答まですべて自動 (年度、問題によってはあとから人手で修正) で行われたが、本研究では問題文の論理式への読み替えや数式処理システムに入力できる形への変換を人間の手で行う。同様の研

究は他にもあるが [4–6], 本研究では問題の種類を限定することで、論理式への変換の考え方や QE ツールの特性が見えやすくなっている。また、通過領域の問題は人間が図形をどう捉えて数式と結びつけるかという面でも興味深い題材である。

本稿では、通過領域の問題を QE ツールを用いて解くための基本的な考え方を示した後に、それを大学入試問題に応用する方法について述べる。応用では「通過領域」であることが明示されていない面積計算への活用も扱う。また、気軽に試せるように一連の処理を関数の形にする。

本研究では数式処理システムは Mathematica, QE ツールは Mathematica の組み込みシンボルである Reduce コマンドを用いる。Mathematica を用いるのは、広く使われていることと、OS ごとのデスクトップ版以外にもインストール不要で使えるクラウドサービス版である Wolfram Cloud [7] やダウンロードが必要だが条件つきで無料で使える Wolfram 言語の実装である Wolfram Engine [8] もあって気軽に使えることなどの理由による。

2 通過領域の問題の解法の基本

通過領域の問題とは、パラメータの変化などに伴って座標平面上 (あるいは座標空間内) の図形が動くとき、その図形が通過する部分 (通過領域) を表す数式やその図示、面積 (体積) などを求める問題である。本章ではパラメータを含む曲線や領域の通過領域を扱う。

* 放送大学教養学部 (情報コース)

2.1 人間の解き方と数式処理システムの解き方

パラメータ t を含む数式で表された曲線 C_t の通過領域の問題を解くとき、通常、次のような方法がとられる。

方法 1: パラメータの変化に伴う曲線の動きから通過領域をとらえ、それをもとに求める

方法 2a: 曲線を表す数式をパラメータ t の方程式とみて、実数解条件に帰着させる

方法 2b: x, y のうち一方の文字を固定して、 t が変化したときの他方の変数の取りうる値の範囲を求める

このうち、方法 2a, 方法 2b はいずれも、ある t についての C_t が (x, y) を通る、すなわち (x, y) を通る C_t が存在するという事実を数式で表してその数式を処理する方法といえる。ここまで「曲線」としたが、曲線以外の図形の場合も同様である。

人間が解く場合は、問題ごとに処理のしやすさや考えやすさを考えて方法を選ぶ。方法 2a または方法 2b で解く場合も、曲線が“動く”、つまり時間の経過の伴って連続的に位置を変える様子をイメージし、それを状況理解や方針立て、あるいは解答が正しいかの判断に役立てるということも多いだろう。一方、数式処理システムにはそれはできないので、問題文に表された内容を数式（論理式）で表し、それを処理することになる。

2.2 数式処理システムによる解法

問題文を一階述語論理式で表すことができれば、それに QE を適用することでもとの論理式と等価で \exists (存在する), \forall (すべての) といった限量子が無い論理式が得られ、問題を解くことができる。したがって、通過領域の問題では通過領域を表す一階述語論理式を求めることが第一歩となる。

方法 2a や方法 2b と同様の考え方により、パラメータを含む数式 $\varphi_0(x, y, t)$ で表される曲線の通過領域は次のように求められる。

定理 2.1. 実数 t に対して、 xy 平面上で図形

$$C: \varphi_0(x, y, t) \quad (1)$$

を考える。

t が条件 $p(t)$ を満たすように動くとき、図形 C が通過する領域を E とすると、 E は一階述語論理式

$$\exists t(p(t) \wedge \varphi_0(x, y, t)) \quad (2)$$

で表される。

証明 $p(t)$ をみたく各 t について、 $\varphi_0(x, y, t)$ をみたく (x, y) の集合を A_t とすると、 $p(t)$ をみたくすべての t についての A_t の和集合を $\bigcup_{t,p(t)} A_t$ として

$$A_t = \{(x, y) \mid \varphi_0(x, y, t)\}, E = \bigcup_{t,p(t)} A_t$$

であるから

$$(x, y) \in E \Leftrightarrow \exists t, p(t)((x, y) \in A_t)$$

$$\Leftrightarrow \exists t(p(t) \wedge \varphi_0(x, y, t))$$

である。すなわち、 E は式 (2) で表される。(証明終)

次のタイプは入試問題等でよくみられる。

系 2.2. 実数 t に対して、 xy 平面上の曲線 C が

$$C: y = f(x, t) \quad (x_0 \leq x \leq x_1) \quad (3)$$

であり、 t が $t_0 \leq t \leq t_1$ の範囲を動くとき、曲線 C が通過する領域 E は

$$p(t) \equiv t_0 \leq t \leq t_1, \quad (4)$$

$$\varphi_0(x, y, t) \equiv y = f(x, t) \wedge x_0 \leq x \leq x_1 \quad (5)$$

として、定理 2.1 の式 (2) で表される。

証明 t の条件は式 (4)、曲線 C を表す式は式 (5) なので、定理 2.1 より定理 2.1 の式 (2) で表される。(証明終)

通過領域 E を表す限量子付きの論理式が得られたら、その論理式に QE を適用することで E を表す限量子のない論理式が得られ、通過領域 E が求められる。Mathematica で解く場合は、式 (2) を `e`, t の条件 $p(t)$ を `condt`, $\varphi_0(x, y, t)$ を `f` とおくと

```
e=Exists[t, condt && f]
```

となり、これに QE ツールである Reduce 関数を用いて

```
Reduce[e, Reals]
```

(あるいは、変数順序を指定して `Reduce[e, {x,y}, Reals]` や `Reduce[e, {y,x}, Reals]`) とすることで E を求めることができる。また、数式処理システムによっては得られた限量子のない論理式を入力とした領域の図示や領域の面積計算も可能である。Mathematica では組み込みシンボルである `RegionPlot`, `Area`, `ImplicitRegion` を用いて実現でき、一連の操作を関数の形にすることで入力の手間を省くこともできる。たとえば、パラメータ t を含む xy 平面上の図形の通過領域 E が一階述語論理式

$$\exists t \varphi(x, y, t)$$

で表されるとき、コード 1 の関数 (`comparea`) は

```
comparea[t,x,y,phi]
```

と入力することで、 E を表す限量子のない論理式、領域の図示、面積のリストを出力する。




図 1: 問題 3.1 出力

```
comparea[paras_,x_,y_,phi_]:=
Module[{lg,fig,ar,x0,x1,y0,y1,reg},
lg=Reduce[Exists[paras,phi],Reals];
reg=ImplicitRegion[lg,{x,y}];
x0=MinValue[x,Element[{x,y},reg]];
x1=MaxValue[x,Element[{x,y},reg]];
y0=MinValue[y,Element[{x,y},reg]];
y1=MaxValue[y,Element[{x,y},reg]];
fig=RegionPlot[lg,{x,x0-0.1,x1+0.1},{y,y0-0.1,
y1+0.1}];
ar=Area[ImplicitRegion[lg,{x,y}]]//
FullSimplify;
Return[{lg,fig,ar}]
]
```

コード 1: comparea 関数

なので, comparea 関数を用いて

```
comparea[t,x,y,0<=t<=1&&0<=x<=1&&y== -x^3+3t^2*
x-2t^3]
```

とすると, 出力結果 (図 1) より, 通過領域の数式

$$\left(0 \leq x \leq \frac{2}{3} \wedge -x^3 + 3x - 2 \leq y \leq 0\right) \vee \left(\frac{2}{3} < x \leq 1 \wedge -x^3 \leq y \leq 0\right) \quad (6)$$

と領域の図, および, 通過領域の面積 $\frac{11}{12}$ がわかる。

系 2.2 を直接用いる形に実装すると (コード 2, kei22 関数), 問題文からの入力がより容易になる。本問では

```
kei22[-x^3+3t^2*x-2t^3,{x,0,1},y,{t,0,1}]
```

により同じ出力が得られる。

```
kei22[fxt_,{x_,x0_,x1_},y_,{t_,t0_,t1_}]:=
comparea[t,x,y,t0<=t<=t1 && x0 <= x<=x1
&& y== fxt]
```

コード 2: kei22 関数

3 大学入試問題の解法への応用

通過領域の面積等を求める問題の多くは, 定理 2.1 が基本であり, これを用いた実装である comparea 関数 (コード 1) を用いて解くことができる。本節ではこれらを応用して大学入試問題を解く方法について述べる。

3.1 通過領域の基本的な問題

2018 年度山口大学 (医・医, 理・数理科) 第 2 問は系 2.2 および comparea 関数 (コード 1) を用いて解くことができる。

問題 3.1 (2018 年度山口大学 (医・医) 第 2 問).

実数 t に対して, xy 平面上で曲線

$$C: y = -x^3 + 3t^2x - 2t^3 \quad (0 \leq x \leq 1)$$

を考える。 t が $0 \leq t \leq 1$ の範囲を動くとき, 曲線 C が通過する領域を図示し, その面積 S を求めなさい。

【解答】 系 2.2 より, 曲線 C の通過領域 E を表す論理式は

$$\exists t(0 \leq t \leq 1 \wedge 0 \leq x \leq 1 \wedge y = -x^3 + 3t^2x - 2t^3)$$

3.2 囲まれた図形の面積

「囲まれた図形の面積」を求める問題は大学入試問題でもよく扱われる題材の 1 つであり, 人間が解くときは曲線を図示し, 曲線どうしや座標軸との交点の座標を求め, 「囲まれた図形」を視覚的にとらえた上で位置関係に注意しながら面積を立式・計算する。ここでは, 2 つのタイプについて, 囲まれた図形を線分の通過領域とみて QE を適用することで, 曲線を図示せずに面積を求める方法を示す。

3.2.1 曲線で囲まれた図形

$C_1: y = f(x)$, $C_2: y = g(x)$ で表される 2 つの曲線で囲まれる図形 (例: 図 2) を表す一階述語論理式は次のように求められる。




図 2: 曲線で囲まれた図形

定理 3.1. $f(x), g(x)$ は実数 x の連続関数とし、曲線 C_1, C_2 を

$$C_1 : y = f(x), \quad C_2 : y = g(x)$$

とする。 $m \leq x \leq M$ において直線 $x = m, x = M$ と C_1, C_2 で囲まれる図形は

$$\exists s(m \leq x \leq M \wedge 0 \leq s \leq 1 \wedge y = (1-s)f(x) + s \cdot g(x)) \quad (7)$$

で表される。

証明 実数 t について、2点 $F_t(t, f(t)), G_t(t, g(t))$ を考えると、 $m \leq x \leq M$ における直線 $x = m, x = M$ と C_1, C_2 の囲む図形は、 t が $m \leq t \leq M$ の範囲を動くとき、線分 $F_t G_t$ の通過する領域と一致する。線分 $F_t G_t$ はパラメータ s を用いて

$$\exists s(0 \leq s \leq 1 \wedge x = t \wedge y = (1-s)f(t) + s \cdot g(t))$$

と表されるので、定理 2.1 より t が $m \leq t \leq M$ の範囲を動くときの通過領域は

$$\exists t(m \leq t \leq M \wedge \exists s(0 \leq s \leq 1 \wedge x = t \wedge y = (1-s)f(t) + s \cdot g(t)))$$

$$\Leftrightarrow \exists t \exists s(m \leq t \leq M \wedge 0 \leq s \leq 1 \wedge x = t$$

$$\wedge y = (1-s)f(t) + s \cdot g(t))$$

$$\Leftrightarrow \exists s(m \leq x \leq M \wedge 0 \leq s \leq 1$$

$$\wedge y = (1-s)f(x) + s \cdot g(x))$$

で表される。(証明終)

x の範囲が指定されていない場合は次がなりたつ。

系 3.2. 2曲線 C_1, C_2 の共有点が有限個のとき、共有点の x 座標の最大値を M 、最小値を m とすると、2曲線 C_1, C_2 の囲む図形は定理 3.1 の式 (7) で表される。

証明 2曲線 C_1, C_2 の共有点の x 座標の最大値が M 、最小値が m のとき、2曲線 C_1, C_2 の囲む図形は直線 $x = m, x = M$ と2曲線 C_1, C_2 の囲む図形にほかならないので、定理 3.1 より、その図形は同じく式 (7) で表

される。(証明終)

$y = f(x), y = g(x)$ と直線 $x = m, x = M$ の囲む図形 (あるいは、単に $y = f(x), y = g(x)$ の囲む図形) の面積を求めるには、定理 3.1 または系 3.2 を用いて得た論理式を利用すればよい。コード 3、コード 4 の関数 `teiri31`, `kei32` はこの処理を Mathematica の関数にしたものであり、`teiri31[f,g,x,y,m,M]`, `kei32[f,g,x,y]` により、数式、領域の図示、面積が自動で求められる (ただし、 f, g は、それぞれ、 $f(x), g(x)$ を表す数式)。

```
teiri31[f_,g_,x_,y_,m_,M_] :=
Module[{s},
Return[comparea[s, x, y, 0 <= s <= 1 && y ==
(1 - s) * f + s * g && m <= x <= M]]
]
```

コード 3: teiri31 関数

```
kei32[f_,g_,x_,y_] :=
Module[{kai, kaimax, kaimin, lg, ar},
kai = x /. Solve[f==g, x, Reals];
kaimax = Max[kai];
kaimin = Min[kai];
Return[teiri31[f, g, x, y, kaimin, kaimax]]
]
```

コード 4: kei32 関数

囲む図形の面積を求める問題では絶対値付きの関数が題材となることも多いが、Mathematica の組み込みシンボルである `Reduce`, `Solve` は絶対値 (Mathematica では `Abs[]`) にも対応しており、2011 年度京都大学 (前期) 理系第 3 問も解くことができる。

問題 3.2 (2011 年度京都大学 (前期) 理系第 3 問).

xy 平面上で、 $y = x$ のグラフと $y = \left| \frac{3}{4}x^2 - 3 \right| - 2$ のグラフによって囲まれる図形の面積を求めよ。

【解答】 `kei32` 関数 (コード 4) を用いて

`kei32[x, Abs[(3/4)x^2-3]-2, x, y]`

とすると、出力結果 (図 3) より、求める面積は $\frac{208}{27}$ とわかる。




図 3: 問題 3.2 の出力

3.2.2 媒介変数で表された曲線で囲まれた図形

媒介変数表示された曲線で囲まれた図形も、線分の通過領域とみて領域を表す論理式を求めることができる。

x 軸に垂直な線分を考えることで次の定理が成り立つ。

定理 3.3. xy 平面上の曲線 C が、媒介変数 t ($t_0 \leq t \leq t_1$) と t の連続関数 $x(t)$, $y(t)$ を用いて

$$x = x(t), y = y(t)$$

と表されているとする。 $t_0 \leq t \leq t_1$ で $x(t)$ は単調増加または単調減少、かつ、つねに $y(t) \geq 0$ で $y(t_0) = y(t_1) = 0$ のとき、 C と x 軸で囲まれる図形（境界を含む）を D とすると、 D は一階述語論理式

$$\exists t(t_0 \leq t \leq t_1 \wedge x = x(t) \wedge 0 \leq y \leq y(t)) \quad (8)$$

で表される。

証明 $P_t(x(t), y(t))$, $H_t(x(t), 0)$ とおくと、 $x(t)$, $y(t)$ が条件をみたすとき線分 P_tH_t は D に含まれるから、 t が $t_0 \leq t \leq t_1$ を動くときの線分 P_tH_t の通過領域は D である。線分 P_tH_t は

$$x = x(t) \wedge 0 \leq y \leq y(t)$$

と表されるので、定理 2.1 より線分 P_tH_t の通過領域、すなわち D は式 (8) で表される。（証明終）

定理 3.3 は図 4 のような領域であれば適用できるが、図 5 のような領域では適用できない。しかし、原点を通る線分を考えることで次の定理 3.4 が成り立ち、図 5 のような領域を表す論理式を求めることができる。




図 4: 定理 3.3 の領域




図 5: 定理 3.4 の領域

定理 3.4. xy 平面上の曲線 C が、媒介変数 t ($t_0 \leq t \leq t_1$) と t の連続関数 $x(t)$, $y(t)$ を用いて

$$x = x(t), y = y(t)$$

と表されているとし、 $x(t)$, $y(t)$ は $t_0 \leq t \leq t_1$ で $y(t)$ は単調増加または単調減少、 $(x(t), y(t))$ が O と異なるとき $x(t) > 0$ とする。

$t = t_0, t_1$ のときの C の点をそれぞれ P_0, P_1 , C と線分 OP_0, OP_1 で囲まれる図形を D とすると、 D は一階述語論理式

$$\exists t \exists s(t_0 \leq t \leq t_1 \wedge 0 \leq s \leq 1 \wedge x = s \cdot x(t) \wedge y = s \cdot y(t)) \quad (9)$$

で表される。

証明 $P_t(x(t), y(t))$ とおくと、 $x(t)$, $y(t)$ が条件をみたすとき、線分 OP_t の通過領域は D である。線分 OP_t 上の点 (x, y) は、実数 s ($0 \leq s \leq 1$) を用いて

$$x = s \cdot x(t) \wedge y = s \cdot y(t)$$

と表されるから、線分 OP_t は

$$\exists s(0 \leq s \leq 1 \wedge x = s \cdot x(t) \wedge y = s \cdot y(t))$$

と表される。したがって、定理 2.1 より t が $x_1 \leq t \leq x_n$ の範囲を動くときの OP_t 通過領域は

$$\exists t(t_0 \leq t \leq t_1 \wedge \exists s(0 \leq s \leq 1 \wedge x = s \cdot x(t)$$

$$\wedge y = s \cdot y(t)))$$

$$\Leftrightarrow \exists t \exists s(t_0 \leq t \leq t_1 \wedge 0 \leq s \leq 1 \wedge x = s \cdot x(t)$$

$$\wedge y = s \cdot y(t))$$

で表される。すなわち、 D は式 (9) で表される。（証明終）

囲まれた図形の図示や面積を求めるには、3.2.1 のときと同様に定理 3.3 や定理 3.4 で得た論理式に QE を適用し、その結果を利用すればよい。コード 5、コード 6 はそれぞれ、定理 3.3、定理 3.4 を利用した Mathematica の関数であり、例題 3.3 のような問題を解くことができる。




図 6: 例題 3.3 の出力

```
teiri33[xt_,yt_,x_,y_,{t_,t0_,t1_}] :=
  comparea[t, x, y, t0<=t<=t1 && x==xt && 0 <=
  y<=yt]
```

コード 5: teiri33 関数

```
teiri34[xt_,yt_,x_,y_,{t_,t0_,t1_}] :=
  comparea[{s,t}, x, y, 0<=s<=1 && t0<=t<=t1
  && x==s*xt && y==s*yt]
```

コード 6: teiri34 関数

例題 3.3. $-1 \leq t \leq 1$ をみたす実数 t に対して

$$x(t) = (1+t)\sqrt{1+t}$$

$$y(t) = 3(1+t)\sqrt{1-t}$$

となる座標平面上の点 $P(x(t), y(t))$ を考え、 t が $-1 \leq t \leq 1$ を動くときの P の軌跡を C とする。 C と x 軸で囲まれた領域 D の面積を求めよ。

【解答】コード 5 の teiri33 関数を用いて

```
teiri33[(1+t)Sqrt[1+t], 3(1+t)Sqrt[1-t], x, y,
  {t,-1,1}]
```

とすると、出力結果 (図 6) より、面積は $\frac{9\pi}{4}$ である (コード 6 の teiri34 関数を用いても同様)。

3.3 回転移動

通過領域の問題の中には回転移動 (図形が形や大きさを変えず、ある定点を中心に一定の角度を回転する) によるものがあり、人間が解く場合、2.1 の方法 1 をとる。すなわち、回転移動による図形の動きをイメージして通過領域の概形をつかみ、計算などにより必要な数値を求めたり状況を確認したりして通過領域を決定する。しかし、数式処理システムを利用して解く場合は、このタイプの問題についても数式で表して処理することになる。

3.3.1 回転移動での通過領域の論理式

xy 平面上の図形 D を、原点中心に反時計回りに θ 回転させるときの通過領域を考える。 D を角 t ($0 \leq t \leq \theta$) だけ回転移動させた図形 D_t を t を用いて表して系 2.2 により通過領域を求める方法もあるが、この場合、 D_t は $\sin t, \cos t$ ($0 \leq t \leq \theta$) を含む式になる。QE で扱うのは多項式からなる一階述語論理式であり、超越関数を含む場合は一般には実行できない [1]。 D_t の式を $\sin t = s, \cos t = c$ とおいて、 $s^2 + c^2 = 1$ をみたす変数 s, c を用いて表すことで多項式で表された問題に帰着させる方法もあるが、ここでは別の方法を考える。

補題 3.5. 原点と異なる点 $P(a, b)$ を原点中心に反時計回りに θ ($-\pi < \theta < \pi, \theta \neq 0$) 回転したときに通過する部分を E とする。このとき、点 $Q(x, y)$ が E 上にあるための必要十分条件は

$$x^2 + y^2 = a^2 + b^2 \wedge (bx - ay) \cdot \sin \theta \leq 0$$

$$\wedge ((a \sin \theta + b \cos \theta)x - (a \cos \theta - b \sin \theta)y) \cdot \sin \theta \geq 0$$

である。とくに、 $\theta = -\frac{\pi}{2}$ のとき、すなわち、時計回りに 90° 回転するとき

$$x^2 + y^2 = a^2 + b^2 \wedge bx - ay \geq 0 \wedge ax + by \geq 0$$

である。

証明 点 $P(a, b)$ を反時計回りに θ 回転した点を $P'(a', b')$ とおくと、 P が通過する部分 E は原点 O を中心とする円弧 $\widehat{PP'}$ なので、 Q が E 上にある条件は、直線 OP を l 、直線 OP' を l' として

$$(OQ = OP)$$

かつ (Q は l 上または l に関して P' と同じ側)
 かつ (Q は l' 上または l' に関して P と同じ側)
 である。

$a' = a \cos \theta - b \sin \theta, b' = a \sin \theta + b \cos \theta$ であり、 l, l' の方程式はそれぞれ、 $bx - ay = 0, b'x - a'y = 0$ なの

で、 x, y のみたすべき条件は

$$\begin{aligned} x^2 + y^2 = a^2 + b^2 \wedge (bx - ay)(ba' - ab') \geq 0 \\ \wedge (b'x - a'y)(b'a - a'b) \geq 0 \\ \Leftrightarrow x^2 + y^2 = a^2 + b^2 \wedge (bx - ay) \sin \theta \leq 0 \\ \wedge ((a \sin \theta + b \cos \theta)x - (a \cos \theta - b \sin \theta)y) \sin \theta \geq 0 \end{aligned}$$

となる。（証明終）




図 7: 点 P の回転（補題 3.5）

定理 3.6. xy 平面上の領域 D が数式 $\varphi_D(x, y)$ で表されているとする。 D を原点を中心として反時計回りに θ ($-\pi < \theta < \pi, \theta \neq 0$) 回転したときに通過する領域を E とするとき、 E は

$$\begin{aligned} \exists a \exists b (\varphi_D(a, b) \wedge \\ x^2 + y^2 = a^2 + b^2 \wedge (bx - ay) \cdot \sin \theta \leq 0 \wedge \\ ((a \sin \theta + b \cos \theta)x - (a \cos \theta - b \sin \theta)y) \cdot \sin \theta \geq 0) \end{aligned} \quad (10)$$

と表すことができる。とくに、 $\theta = -\frac{\pi}{2}$ のとき、すなわち、時計回りに 90° 回転するとき、 E は

$$\begin{aligned} \exists a \exists b (\varphi_D(a, b) \wedge x^2 + y^2 = a^2 + b^2 \wedge bx - ay \geq 0 \\ \wedge ax + by \geq 0) \end{aligned} \quad (11)$$

と表すことができる。

証明 $P \in D$ が通過する点の集合を B_P とすると、 $E = \bigcup_{P \in D} B_P$ だから、

$$(x, y) \in E \Leftrightarrow \exists P (P \in D) ((x, y) \in B_P)$$

である。ここで、 $P(a, b)$ が D の点である条件は $\varphi_D(a, b)$ である。また、補題 3.5 より、 (x, y) が $P(a, b)$ の通過領域 B_P に含まれる条件 $\varphi_P(x, y)$ は

$$\begin{aligned} \varphi_P(x, y) \equiv x^2 + y^2 = a^2 + b^2 \wedge (bx - ay) \sin \theta \leq 0 \wedge \\ ((a \sin \theta + b \cos \theta)x - (a \cos \theta - b \sin \theta)y) \sin \theta \geq 0 \end{aligned}$$

である。これにより通過領域 E は式 (10) のように表される。（証明終）

θ は定数なので、定理 3.6 の中に出てくる $\sin \theta, \cos \theta$ は定数である。したがって、 $\sin \theta, \cos \theta$ が代数的数であれば定理 3.6 の式に QE を用いることができる。

（補足） $\pi \leq |\theta| < 2\pi$ の場合は、 $\theta = \theta_1 + \theta_2$ ($|\theta_1| < \pi, |\theta_2| < \pi$) として求めればよい (D を θ_1 回転したとき

の通過領域を E_1, E_1 を θ_2 回転したときの通過領域を E_2 とすると、 $E = E_1 \cup E_2$ である)。なお、 $|\theta| \geq 2\pi$ の場合（1 回転以上する場合）の通過領域 E は

$$\exists a \exists b (\varphi_D(a, b) \wedge x^2 + y^2 = a^2 + b^2)$$

と表せる。

定理 3.6 を利用して平面上の図形 D を回転したときの通過領域の論理式と面積を求める Mathematica の関数 (teiri36) をコード 7 に示す (teiri36[d, theta, x, y] の入力 d は D を表す限量子のない論理式、theta は回転角（弧度法、反時計回りを正）、x, y は変数であり、論理式と面積のリストを出力とする）。

```
teiri36[d_, theta_, x_, y_] :=
Module[{d1, a, b, sinT, cosT},
d1=d/.{x->a, y->b};
sinT=Sin[theta];
cosT=Cos[theta];
Return[comparea[{a,b}, x, y, d1 && x^2+y^2==a^2+b^2 && sinT*(b*x-a*y)<= 0 && sinT*((a*
sinT + b*cosT)*x-(a*cosT-b*sinT)*y ) >= 0]]
]
```

コード 7: teiri36 関数

次の問題は、人間が解く場合は方法 1（図形の動きから考える）で解くが、数式処理システムで解くことができる。

例題 3.4. $A(1, 0), B(2, 0), C(2, 1)$ を頂点とする三角形 ABC の周および内部を、原点 $O(0, 0)$ を中心に反時計回りに 60° 回転させるとき、三角形 ABC が通過する領域の面積を求めよ。




図 8: 三角形の回転（例題 3.4）

【解答】 回転角は 60° 、すなわち $\frac{\pi}{3}$ で、三角形 ABC の周および内部は

$$\varphi_D(x, y) \equiv y \geq 0 \wedge x \leq 2 \wedge y \leq x - 1$$

と表されるので、コード 7 の teiri36 関数を用いて計算すると（図 9）、求める面積は $\frac{1}{6}(3 + 4\pi)$ とわかる。

```
In[5]:= teiri36[y >= 0 && x <= 2 && y <= x - 1, Pi/3, x, y][[3]];
FullSimplify [%]
Out[6]= 1/6 (3 + 4 Pi)
```

図 9: 例題 3.4 の入出力

(補足) teiri36 は数式, 図, 面積を要素とするリストを返すので, [[3]] により 3 つ目の要素を取り出した。

3.3.2 パラメータ表示された曲線の回転

回転前の図形 D の境界がパラメータ (ただし, $x(t)$, $y(t)$ が条件をみたすもの) を用いて表されているときは, D の論理式に定理 3.3 または定理 3.4 のときの考え方が利用できる。ただし, この節では, 回転前の図形 D は境界を含まないものとする。

(補足) QE では等号付きの不等式 $f(x, y) \geq 0$ は

$$f(x, y) \geq 0 \Leftrightarrow f(x, y) > 0 \vee f(x, y) = 0$$

として扱うため, 等号なしの場合と比べて計算量が大きく, 計算に時間がかかる。面積や領域の概形をつかむだけなら内点のみを考えたので十分な場合も多いので, 境界を含めないこととした。QE では問題を小さな問題に分けて処理する方が計算量が小さくなる傾向があり [1], 内点以外の点も考える必要がある場合も, 時間がかかるようなら内点とそれ以外を別々に処理して後でまとめてみるとよい。

定理 3.3 の考え方をを用いる場合, 回転前の図形 D (境界を含まない) の論理式は

$$\exists t(t_0 < t < t_1 \wedge x = x(t) \wedge 0 < y < y(t))$$

なので, 定理 3.6 より D の通過領域 E を表す論理式は, $k(a, b)$ を

$$k(a, b)$$

$$\equiv x^2 + y^2 = a^2 + b^2 \wedge (bx - ay) \cdot \sin \theta < 0 \wedge$$

$$((a \sin \theta + b \cos \theta)x - (a \cos \theta - b \sin \theta)y) \cdot \sin \theta < 0$$

として

$$\begin{aligned} \exists a \exists b (\exists t(t_0 < t < t_1 \wedge x = x(t) \wedge 0 < y < y(t)) \\ \wedge k(a, b)) \end{aligned}$$

$$\Leftrightarrow \exists a \exists b \exists t(t_0 < t < t_1 \wedge x = x(t) \wedge 0 < y < y(t) \wedge k(a, b))$$

となる。これに QE を適用すれば通過領域 E を表す限量子のない論理式が得られる。コード 9 に, D の境界の曲線のパラメータ表示と回転角から E の論理式を求める関数 (kaiten33) を示す。ただし, コード 9 および後述のコード 10 で用いる complgfig 関数はコード 8 によるものである (complgfig 関数はコード 1 の comparea 関数の面積計算を省略したもの)。

定理 3.4 を用いる場合も同様に考えると, 回転前の図形 D の論理式は

$$\exists t \exists s(t_0 < t < t_1 \wedge 0 < s < 1 \wedge x = s \cdot x(t) \wedge y = s \cdot y(t))$$

なので, 定理 3.6 より D の内点の通過領域 E を表す論

理式は

$$\begin{aligned} \exists a \exists b ((\exists t \exists s(t_0 < t < t_1 \wedge 0 < s < 1 \wedge x = s \cdot x(t) \\ \wedge y = s \cdot y(t)) \wedge k(a, b)) \\ \Leftrightarrow \exists a \exists b \exists t \exists s(t_0 < t < t_1 \wedge 0 < s < 1 \wedge x = s \cdot x(t) \\ \wedge y = s \cdot y(t) \wedge k(a, b)) \end{aligned}$$

となる。コード 10 に, この場合の D の境界の曲線のパラメータ表示と回転角から E の論理式を求める関数 (kaiten34) を示す。

```
complgfig[paras_,x_,y_,phi_]:=
Module[{lg,fig,ar,x0,x1,y0,y1,reg},
lg=Reduce[Exists[paras,phi],Reals];
reg=ImplicitRegion[lg,{x,y}];
x0=MinValue[x,Element[{x,y},reg]];
x1=MaxValue[x,Element[{x,y},reg]];
y0=MinValue[y,Element[{x,y},reg]];
y1=MaxValue[y,Element[{x,y},reg]];
fig=RegionPlot[lg,{x,x0-0.1,x1+0.1},{y,y0-0.1,
y1+0.1}];
Return[{lg,fig}]
]
```

コード 8: complgfig 関数

```
kaiten33[{xt_,yt_},{t_,t0_,t1_},theta_,x_,y_]:=
Module[{a,b,sinT,cosT},
sinT=Sin[theta];
cosT=Cos[theta];
Return[complgfig[{a,b,t},x,y,t0<t<t1 && a
==xt && 0<b<yt && x^2+y^2==a^2+b^2 && sinT*(
b*x-a*y)<0 && sinT*((a*sinT+b*cosT)*x-(a*
cosT-b*sinT)*y)>0]]
]
```

コード 9: kaiten33 関数

```
kaiten34[{xt_,yt_},{t_,t0_,t1_},theta_,x_,y_]:=
Module[{a,b,s,sinT,cosT},
sinT=Sin[theta];
cosT=Cos[theta];
Return[complgfig[{a,b,s,t},x,y,t0<t<t1
&& a>0 && 0<s<1 && a==s*xt && b==s*yt && x
^2+y^2==a^2+b^2 && sinT*(b*x-a*y)<0 && sinT
*((a*sinT+b*cosT)*x-(a*cosT-b*sinT)*y)>0]]
]
```

コード 10: kaiten34 関数

これを用いて 2020 年東京大学 (前期) 理系第 3 問の解答を試みる。受験生が解く場合は, (1), (2) の内容を含む考察によって領域 D の特徴を式と図の両方から捉えた後, 通過領域の図形を求めて面積を計算することになり, (2) が特に重要であるが, 数式処理システムでは (2) は不要である。




図 10: 問題 3.5 の出力

問題 3.5 (2020 年東京大学理学系第 3 問).

$-1 \leq t \leq 1$ をみたす実数 t に対して

$$x(t) = (1+t)\sqrt{1+t}$$

$$y(t) = 3(1+t)\sqrt{1-t}$$

とする。座標平面上の点 $P(x(t), y(t))$ を考える。

- (1) $-1 \leq t \leq 1$ における t の関数 $\frac{y(t)}{x(t)}$ は単調に減少することを示せ。
- (2) 原点と P の距離を $f(t)$ とする。 $-1 \leq t \leq 1$ における t の関数 $f(t)$ の増減を調べ、最大値を求めよ。
- (3) t が $-1 \leq t \leq 1$ を動くときの P の軌跡を C とし、 C と x 軸で囲まれた領域を D とする。原点を中心として D を時計回りに 90° 回転させるとき、 D が通過する領域の面積を求めよ。

【解答】 $x(t), y(t)$ の式および (1) より、定理 3.3 の条件をみたすので、コード 9 の関数 (kaiten33) を用いて

```
xt = (1+t) Sqrt [1+t];
yt = 3 (1+t) Sqrt [1-t];
ansp = kaiten33[{xt, yt},{t,-1,1},-Pi/2,x,y]
```

とすると、図 10 のように出力された。(1) より定理 3.4 の条件もみたすのでコード 10 の関数 (kaiten34) を用い

てもよい。この場合も同じ出力となる)

出力された論理式を用いて本稿の他の問題と同様に面積計算を試みたが、計算結果は得られなかった。

4 活用と注意

4.1 学習や問題作成への活用

前節で示した定理や関数を活用することで通過領域の面積や囲まれる領域の面積を簡単に解くことができる。入試問題などの既存の問題を解くだけでなく、関数の式やパラメータの変域などの条件を少し変えた場合の領域の概形や面積を求めるのも容易なので、条件や式の形が少し違う場合を調べるときや、類題の作成、解答の確認などに役立てることができる。人間が解く場合、視覚的にとらえた図形の情報や図形の動きのイメージが役立つが、図形の位置関係の誤った思い込みによるミスも起こりがちである。手計算の結果とシステムによる計算結果を比べることで出題ミスの予防につなげることが可能である。

4.2 注意点

QE および数式処理システムは大変有用であるが、QE には計算量が大きいという欠点があり、実際は正しく立

式・入力しても現実的な時間ではQEの計算が終わらない場合がある。入力の工夫により計算時間が短くなる場合があるので、うまくいかない場合は小問題への分割など、入力を工夫してみるとよい。計算できても出力の数式が複雑で人間が直接図示等に役立てるのが難しい場合もある。また、問題3.5で見たように、QEの計算や図示はできても面積はできないという場合もある(Mathematicaは商用ソフトであり、面積計算で用いる組み込みシンボルについての詳細はわからない)。システムで計算できなかったり時間がかかったりすることと、人間が手計算で解く場合の難しさは別なので注意が必要である。

5 まとめ

本稿で与えた定理と数式処理システムを活用することにより、通過領域の問題を解くことができた。今回示した関数への入力は問題文から読み取れる内容であり、問題文の条件や変数の意味の理解につながる事が期待できる。面倒な処理なしに多数の問題を試せることも利点であり、ある問題についてその式や条件を変えながら解くことにより、その問題の本質的な性質を探ることもできる。

QEは強力なツールであり、うまく使いこなすことで解ける問題が増える。そのためには一階の論理式への読み替えが必要であり、計算を可能にするために入力を工夫しなければならないこともあるが、そのことにはまた、新たな気づきを生むことになるだろう。

人間が図形の問題を解く場合、図形をひとつのまとまりとして視覚的にとらえたり、動きをイメージしたりするが、数式処理システムを利用する場合はそれらをすべて数式として表現する必要がある、そのためには見て、あるいは、イメージで「わかる」ことは何なのかを改めて問い直すことになる。ツールの利用や関数の作成を通じた新たな視点の発見や能力の向上にも期待したい。

参考文献

- [1] 穴井宏和, 横山和弘 (2011) 『QEの計算アルゴリズムとその応用』東京大学出版会
- [2] 松崎 拓也 (2017) 「数学問題の自然言語解析」『科学哲学』50, pp.35-49
- [3] 新井紀子, 東中竜一郎 (2018) 『人工知能プロジェクト「ロボットは東大に入れるか」第三次 AI ブームの到達点と限界』東京大学出版会
- [4] 矢吹太郎 (2014) 「数学入試問題における数式処理システムの性能評価」*The 28th Annual Conference of Japanese Society for Artificial Intelligence*
- [5] 矢吹太郎 (2017) 「機械まかせの数学—Mathematicaで解く東大入試数学」『大学への数学』2017年6月号(第61巻第5号), pp.66-69, 東京出版
- [6] 森谷慧士, 矢吹太郎 (2016) 「大学入試センター試験数学I・Aを用いた数式処理システムの性能評価」第78回全国大会講演論文集第2016巻1号 pp.849-850
- [7] WOLFRAM CLOUD,
<https://www.wolframcloud.com/>
- [8] Free Wolfram Engine for Developers,
<https://www.wolfram.com/engine/>